Fórmula de Taylor

El objetivo de esta práctica es aproximar funciones de una variable real por polinomios, mediante la fórmula de Taylor.

1. Fórmula de Taylor

1) Escribir $x^4 + x^3 - 3x^2 + 4x - 4$ en potencias de $(x-1)$

La orden **taylor** proporciona el desarrollo de Taylor de una función en un punto, del orden que se quiera. La orden **taylor** se puede sustituir por **series**.

En nuestro caso, queremos el desarrollo del polinomio en el punto 1, de orden igual al grado del polinomio. Es decir, de orden 4 (obsérvese que se escribe **taylor(...,5)**).

```
> taylor(x^4+x^3-3*x^2+4*x-4,x=1,5);
```

$$-1 + 5(x-1) + 6(x-1)^2 + 5(x-1)^3 + (x-1)^4$$

2) Hallar un polinomio que aproxime $f(x) = e^x$ en el intervalo $[-1, 1]$ con error menor que 10^{-2}.

Empezamos de nuevo, definimos la función, la representamos en el intervalo $[-1, 1]$ y ponemos un título. También cargamos un "paquete" de Maple que nos permite utilizar más adelante algunas órdenes adicionales. El paquete se llama **plots** y la orden para cargarlo es **with**.

```
> restart;f:=x->exp(x):with(plots):
  plot(f(x),x=-1..1,title="La función exp(x)");
```



Con la orden **taylor** obtenemos su desarrollo de Taylor en el punto y del orden que queramos. Por ejemplo, en $x = 0$ y de orden 4 (de nuevo, obsérvese que se escribe **taylor(...,5)**):

```
> taylor(f(x),x=0,5);
```

$$1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4 + O(x^5)$$

El último término es una O "grande" de x^5 y representa el resto de la fórmula de Taylor. Si lo que queremos es quedarnos con el polinomio, usamos la orden **convert(..., polynom)**:

```
> convert(%,polynom);
```

$$1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4$$

Por ejemplo, el polinomio de orden 2 se obtiene así:

```
> convert(taylor(f(x),x=0,3),polynom);
```

$$1 + x + \frac{1}{2}x^2$$

Para no complicar la notación, podemos dar un nombre al polinomio de grado 2:

```
> pol:=x->convert(taylor(f(x),x=0,3),polynom);
```

$$polinomio := x \to convert(taylor(f(x), x = 0, 3), polynom)$$

Representemos la función f, de color rojo, y el polinomio de grado 2, de color azul:

```
> plot([f(x),pol(x)],x=-1..1,color=[red,blue]);
```



¿Con qué error aproxima este polinomio a la función? El error es el valor absoluto de la diferencia. En la gráfica anterior no se ve bien de qué tamaño es este error, así que vamos a representarlo:

```
> plot(abs(f(x)-pol(x)),x=-1..1);
```



Nos piden que el error sea menor que 10^{-2} y en este caso claramente no lo es. Por si hubiera alguna duda, repetimos la gráfica añadiendo la recta $y = 10^{-2}$:

```
> plot([abs(f(x)-pol(x)),10^(-2)],x=-1..1,color=[red,blue]);
```



Ahora podemos tantear con polinomios de mayor grado, hasta que la gráfica del error (en color rojo) quede por debajo de la recta horizontal en todo el intervalo $[-1, 1]$. Empezamos con el polinomio de grado 3:

```
> pol:=x->convert(taylor(f(x),x=0,4),polynom):
  plot([abs(f(x)-pol(x)),10^(-2)],x=-1..1,color=[red,blue]);
```



Tampoco sirve. El de grado 4 (sería más cómodo escribir sobre la orden anterior):

```
> pol:=x->convert(taylor(f(x),x=0,5),polynom):
  plot([abs(f(x)-pol(x)),10^(-2)],x=-1..1,color=[red,blue]);
```



No está muy claro si sirve o no. Dibujamos la gráfica cerca de 1:

```
> plot([abs(f(x)-pol(x)),10^(-2)],x=0.9..1,color=[red,blue]);
```



Si Maple no está equivocado, el polinomio de grado 4 resuelve nuestro problema:

```
> pol(x);
```

$$1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4$$

Deberíamos probar con el rigor debido que esa es la solución. Pero eso lo hacemos en papel.